A

EXERCISES OF SELF-ASSESSMENT

▼ Proposed Exercise A-1

a) Define the following two functions: f(x,y) = sin(x)sin(y) - 0.5 and g(x,y) = cos(x)cos(y) - 0.5.
b) Do the graphical representation of the curves f(x,y) =0 and g(x,y) =0 using the same axis, use different colours in the representation of each of the graphs and give colour to the back of the graphic .

▼ Resolution A-1

a) Definition of the functions

```
f[x_, y_] = Sin[x] * Sin[y] - 0.5;
g[x_, y_] = Cos[x] * Cos[y] - 0.5;
```

b) Graphical representation of the functions

```
ContourPlot[{f[x, y] == 0, g[x, y] == 0}, {x, -2 π, 2 π}, {y, -2 π, 2 π},
 ContourStyle → {{Thickness[0.01], Yellow}, {Thickness[0.01], Green}},
 Axes → True, AxesLabel → {"OX", "OY"}, Background → LightBlue]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
ContourPlot[{f[x, y]}, {x, -2 π, 2 π}, {y, -2 π, 2 π},
 ContourStyle → {{Thickness[0.01], Yellow}, {Thickness[0.01], Green}},
 Axes → True, AxesLabel → {"OX", "OY"}, Background → LightBlue]
```



▼ Proposed Exercise A-2

Make the graphical representation of the following family of roses:



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

▼ Resolution A-2

```
g1 = PolarPlot[Sin[4 θ], {θ, 0, 2 Pi}, AxesLabel → {x, y}, Mesh → 15,
    MeshFunctions → {#1 &}, MeshStyle → Red, PlotLabel → r == Sin[4 θ]];
g2 = PolarPlot[Sin[6 θ], {θ, 0, 2 Pi}, AxesLabel → {x, y}, PlotStyle → Thick,
    Mesh → 15, MeshShading → {Orange, Green}, PlotLabel → r == Sin[6 θ]];
g3 = PolarPlot[Cos[6 θ], {θ, 0, 2 Pi}, AxesLabel → {x, y}, PlotStyle → Thick,
    ColorFunction → Function[{x, y, θ}, Hue[θ / (4 Pi)]],
    ColorFunctionScaling → False, PlotLabel → r == Sin[6 θ]];
g4 = PolarPlot[{4 + 3 * Sin[12 * (t - 0.1)], 4 + 3 * Cos[12 * t], 1, 7}, {t, 0, 2 π},
   ColorFunction → "DarkRainbow", PlotStyle → Directive[Red, Thick]];
GraphicsGrid[{{g1, g2}, {g3, g4}}]
```



▼ Proposed Exercise A-3

Make the graphical representation of the following family of cardioids:

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez



▼ Resolution A-3

Cardioid 1

```
cardioid1[t_, a_] = a (1 + Cos[t]);
```

Cardioid 2

```
cardioide[t_, a_] = a (1 - Cos[t]);
```

Cardioid 3

```
cardioid3[t_, a_] = a (1 + Sin[t]);
```

Cardiod 4

```
cardioid4[t_, a_] = a (1 - Sin[t]);
```

Family of cardioids

```
c1 = PolarPlot[Evaluate[Table[cardioid1[t, a], {a, 1, 10, 2}]], {t, 0, 2 Pi}, PlotStyle ->
    Table[RGBColor[0, 0.5, i * 0.1], {i, 1, 10, 2}], PlotLabel -> r == a (1 + Cos[t])];
c2 = PolarPlot[Evaluate[Table[cardioid2[t, a], {a, 1, 10, 2}]], {t, 0, 2 Pi},
  PlotStyle -> Table[RGBColor[i * 0.1, 0, 0.5], {i, 1, 10, 2}],
  PlotLabel -> r == a (1 - Cos[t])];
c3 = PolarPlot[Evaluate[Table[cardioid3[t, a], {a, 1, 10, 2}]], {t, 0, 2 Pi}, PlotStyle ->
   Table[RGBColor[0.2, 0, i * 0.1], {i, 1, 10, 2}], PlotLabel -> r == a (1 + Sin[t])];
c4 = PolarPlot[Evaluate[Table[cardioid4[t, a], {a, 1, 10, 2}]], {t, 0, 2 Pi},
  PlotStyle -> Table[RGBColor[0.5, i * 0.1, 0], {i, 1, 10, 2}],
  PlotLabel -> r == a (1 - Sin[t])]; GraphicsGrid[{{c1, c2}, {c3, c4}}]
```



▼ Proposed Exercise A-4

Plot the family of lemniscates:



▼ Resolution A-4

```
eqlemn = (x^4) == a^2 (x^2 - y^2)
```

$x^4 == a^2 (x^2 - y^2)$

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
eqlemn = (x^4) == a^2 (x^2 - y^2) /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$a^2 \, Cos[2\,t]\, r[t] == Cos[t]^4\, r[t]^3$

```
lemniscates[t_, a_] = a (Cos[2 * t]) ^ (1 / 2) / Cos[t]^2
```

$a \sqrt{Cos[2\,t]}\; Sec[t]^2$

```
PolarPlot[Evaluate[Table[lemniscates[t, a], {a, 1, 9, 2}]], {t, 0, 2 π},
 PlotLabel → Style["GERONO LEMNISCATES FAMILY", Bold, 14], Background → GrayLevel[0.8]]
```



▼ Proposed Exercise A-5

Giving two functions, define the tangent line to them in any point. Plot the functions and their tangent lines in an interval that contains the point.

▼ Resolution A-5

```
f[x_] = 2 * x / (x^2 + 1);
g[x_] = Cos[x];
x0 = 1 / 2;

tangentf[x_] = f[x0] + f'[x0] * (x - x0);
tangentg[x_] = g[x0] + g'[x0] * (x - x0);

Plot[{f[x], tangentf[x], g[x], tangentg[x]}, {x, -1, 2},
 PlotStyle → {{Blue, Thickness[0.01]}, {Orange, Thickness[0.005], Dashing[0.02]},
   {Green, Thickness[0.01]}, {Purple, Thickness[0.005], Dashing[0.02]}}]
```



▼ Proposed Exercise A-6

Make the animation of the next graphic:



▼ Resolution *A-6*

```
ek = (x - a) ^2 + (y - b) ^2 == c^2
```

$(-a+x)^2 + (-b+y)^2 == c^2$

```
ek3 = ek /. {a → 0, b → 0}
```

$x^2 + y^2 == c^2$

```
polar3 = ek3 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$c^2 == r[t]^2$

```
Solve[polar3, r[t]]
```

{{r[t] → -c}, {r[t] → c}}

```
circ3[t_, c_] = c;
```


E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Manipulate[PolarPlot[{circ3[t, 4], circ3[t, c]}, {t, 0, 2 * π}, PlotStyle →
    {{RGBColor[0.5, 1, 0], Thickness[0.015]}, RGBColor[0.5, 0, 1]}], {c, 1, 4, 0.2}]
```



▼ Proposed Exercise A-7

Make the animation of the next graphic:



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

▼ Resolution A-7

```
Manipulate[PolarPlot[Evaluate[Table[circ3[t, 4 - p], {p, 0, c, 0.2}]], {t, 0, 2 * π}],
 {c, 0.2, 4, 0.2}]
```



▼ Proposed Exercise A-8

Make the animation of the next graphic:



▼ Resolution A-8

```
ek1 = ek /. {a → 0, c → b}
```

$x^2 + (-b + y)^2 == b^2$

```
polar1 = ek1 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

r[t]^2 == 2 b r[t] Sin[t]

```
Solve[polar1, r[t]]
```

{{r[t] → 0}, {r[t] → 2 b Sin[t]}}

```
circ1[t_, b_] = 2 * b Sin[t];
Manipulate[PolarPlot[circ1[t, 1], {t, 0, p}, PlotStyle → RGBColor[0.5, 0, 1]], {p, 0.1, π}]
```



▼ Proposed Exercise A-9

Make the animation of the next graphic:



▼ Resolution A-9

```
ek2 = ek /. {b → 0, c → a}
```

(− a + x)^2 + y^2 == a^2

```
polar2 = ek2 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

2 a Cos[t] r[t] == r[t]^2

2 a Cos[t] r[t] == r[t]^2

2 a Cos[t] r[t] == r[t]^2

```
Solve[polar2, r[t]]
```

{{r[t] → 0}, {r[t] → 2 a Cos[t]}}

```
circ2[t_, a_] = 2 * a Cos[t];
```


E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Manipulate[PolarPlot[circ2[t, 1], {t, 0, p}, PlotStyle → RGBColor[0.5, 0, 1]], {p, 0.1, π}]
```



▼ Proposed Exercise A-10

Make the animation of the next graphic:



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

▼ Resolution A-10

```
Manipulate[PolarPlot[Evaluate[Table[circ2[t, 4 - p], {p, 0, c, 0.2}]], {t, 0, 2 * π}],
 {c, 0.2, 4, 0.2}]
```



▼ Proposed Exercise A-11

Make the animation of the next graphic:



▼ Resolution A-11

```
Manipulate[
 PolarPlot[Evaluate[Table[circ1[t, 4 - p], {p, 0, c, 0.2}]], {t, 0, 2 * π}], {c, 0, 4, 0.2}]
```



▼ Proposed Exercise A-12

a) Study the existence of the repeated limits and radial limits of the function f(x,y)=$\frac{x^2-y^2}{x^2+y^2}$.

b) Study the existence of the repeated limits and radial limits of the function f(x,y)=$\frac{x\,y}{x^2+y^4}$.

▼ Resolution A-12

Section a)

```
f[x_, y_] = (x^2 - y^2) / (x^2 + y^2)
```

$$\frac{x^2 - y^2}{x^2 + y^2}$$

Repeated limits

```
l1 = Limit[Limit[f[x, y], x -> 0], y -> 0]
```

-1

```
l2 = Limit[Limit[f[x, y], y -> 0], x -> 0]
```

1

Directional limits

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Limit[f[x, m * x], x -> 0]
```

$$\frac{1 - m^2}{1 + m^2}$$

The radial limits do not exist

```
Plot3D[f[x, y], {x, -4, 4}, {y, -4, 4}, Mesh → False]
```




```
ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}]
```



Section b)

```
f[x_, y_] = (x * y) / (x^2 + y^4)
```

$$\frac{x\, y}{x^2 + y^4}$$

Repeated limits

The marginal funtion f2[y] does not exist when x→0

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
l1 = Limit[Limit[f[x, y], x -> 0], y -> 0]
0
l2 = Limit[Limit[f[x, y], y -> 0], x -> 0]
0
```

Directional limits

```
Limit[f[x, m * x], x -> 0]
m
```

The radial limits do not exist

```
Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}, Mesh → False]
```




```
ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}]
```



▼ Proposed Exercise A-13

Given the family of circles that in the point (0,0) are tangent to the OY axis $x^2 + y^2 = cx$,

a) Obtain its Differential Equation and solve it.

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

b) Obtain a family of solutions and plot it.
c) Obtain the Differential Equation of the orthogonal trajectory and solve it.
d) Obtain a family of solutions and plot it.
e) Plot the families of both curves and the vectorial fields of each of them together with the orthogonal trajectories.

Resolution A-13

Section a)

- **Equation of the family of curves $x^2 + y^2 = c * x$**

```
ek = x^2 + y[x]^2 == c * x
```

$x^2 + y[x]^2 == c\,x$

```
con = Solve[ek, c]
```

$\left\{\left\{c \to \frac{x^2 + y[x]^2}{x}\right\}\right\}$

```
ed = D[ek, x] /. c → con[[1, 1, 2]]
```

$2\,x + 2\,y[x]\,y'[x] == \frac{x^2 + y[x]^2}{x}$

Section b)

- **Resolution of the Differential Equation of the family of curves**

```
Si = DSolve[ed, y[x], x]
```

$\left\{\left\{y[x] \to -\sqrt{-x^2 + x\,C[1]}\right\}, \left\{y[x] \to \sqrt{-x^2 + x\,C[1]}\right\}\right\}$

```
s1[x_, c_] = Si[[1, 1, 2]] /. C[1] → c / 2
s2[x_, c_] = Si[[2, 1, 2]] /. C[1] → c / 2
```

$-\sqrt{\frac{c\,x}{2} - x^2}$

$\sqrt{\frac{c\,x}{2} - x^2}$

- **A family of solutions is given by the following list**

```
solti = Table[{s1[x, c], s2[x, c]}, {c, -20, 20., 2}];

listasolti = Flatten[solti, 2];

famsolti = Plot[Evaluate[listasolti], {x, -10, 10}, PlotStyle →
   {{Orange, Thickness[0.004]}, {Orange, Thickness[0.004]}}, AspectRatio → Automatic]
```



- **In each point (x,y), the directional vector of the tangent line to the curve is (1,m), where**

```
m = Solve[ed, y'[x]]
```

$$\left\{\left\{y'[x] \to \frac{-x^2 + y[x]^2}{2\,x\,y[x]}\right\}\right\}$$

Section c)

- **Differential Equation of the orthogonal trajectories**

```
edto = y'[x] == -1 / m[[1, 1, 2]]
```

$$y'[x] == -\frac{2\,x\,y[x]}{-x^2 + y[x]^2}$$

```
So = DSolve[edto, y[x], x]
```

$$\left\{\left\{y[x] \to \frac{1}{2}\left(e^{C[1]} - \sqrt{e^{2\,C[1]} - 4\,x^2}\right)\right\}, \left\{y[x] \to \frac{1}{2}\left(e^{C[1]} + \sqrt{e^{2\,C[1]} - 4\,x^2}\right)\right\}\right\}$$

```
so1[x_, c_] = So[[1, 1, 2]] /. {e^C[1] → c, e^(2*C[1]) → c^2}
so2[x_, c_] = So[[2, 1, 2]] /. {e^C[1] → c, e^(2*C[1]) → c^2}
```

$$\frac{1}{2}\left(c - \sqrt{c^2 - 4\,x^2}\right)$$

$$\frac{1}{2}\left(c + \sqrt{c^2 - 4\,x^2}\right)$$

Section d)

- **A family of solutions is given by the following list**

```
solto = Table[{so1[x, c], so2[x, c]}, {c, -10, 10., 1}];

listasolto = Flatten[solto, 2];

famsolto = Plot[Evaluate[listasolto], {x, -5, 5}, PlotStyle →
   {{Blue, Thickness[0.004]}, {Blue, Thickness[0.004]}}, AspectRatio → Automatic]
```




```
s[x_, c_] = So[[1, 1, 2]] /. C[1] → c
```

$$\frac{1}{2}\left(e^{c} - \sqrt{e^{2\,c} - 4\,x^2}\right)$$

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

Section e)

- **We will plot the family of curves and the orthogonal trajectories**

```
j = Show[{famsolto, famsolti},
  AspectRatio → Automatic, PlotRange → {{-10, 10}, {-10, 10}}]
```



- **We will plot the vectorial fields of the family of curves and their orthogonal trajectories**

```
m
```

$$\left\{\left\{y'[x] \to \frac{-x^2 + y[x]^2}{2\, x\, y[x]}\right\}\right\}$$

```
k = VectorPlot[{{2 * x * y, y^2 - x^2}, {-y^2 + x^2, 2 * x * y}},
  {x, -10, 10}, {y, -10, 10}, VectorPoints → 20, VectorScale → Small,
  StreamScale → Full, VectorStyle → {Purple, Blue}]
```




```
Show[j, k]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez